5. Some Problems in Heaven

Unfortunately, $\dim\mathcal{A}(\mathcal{X},X)=\dim\Lambda(\mathcal{X},X)=4^{|X|}$ is big. Fortunately, we have the following theorem, a version of one of the main results in Halacheva's thesis, [Ha1, Ha2]:

Theorem. Working in $\Lambda(\mathcal{X}\cup X)$, if $w=\omega\mathbb{e}^\lambda$ is a balanced Gaussian (namely, a scalar ω times the exponential of a quadratic $\lambda=\sum_{\zeta\in\mathcal{X},z\in X}\alpha_{\zeta,z}\zeta z$), then generically so is $c_{x,\xi}\mathbb{e}^\lambda$.

(This is great news! The space of balanced quadratics is only $|\mathcal{X}||X|$-dimensional!)

Γ-calculus.

Thus we have an almost-always-defined "Γ-calculus": a contraction algebra morphism $\mathcal{T}(\mathcal{X},X)\to R\times(\mathcal{X}\otimes_{R/R}X)$ whose behaviour under contractions is given by

$$c_{x,\xi}(\omega,\lambda=\mu+\eta x+\xi y+\alpha\xi x)=((1-\alpha)\omega,\mu+\eta y/(1-\alpha)).$$

(Γ is fully defined on pure tangles – tangles without closed components – and hence on long knots).

Multiplying and comparing Γ objects:

```
Γ /: Γ[is1_, os1_, cs1_, ω1_, λ1_] × Γ[is2_, os2_, cs2_, ω2_, λ2_] :=
 Γ[is1 ⋃ is2, os1 ⋃ os2, Join[cs1, cs2], ω1 ω2, λ1 + λ2]
Γ /: Γ[is1_, os1_, _, ω1_, λ1_] ≡ Γ[is2_, os2_, _, ω2_, λ2_] :=
 TrueQ[(Sort@is1 === Sort@is2) ∧ (Sort@os1 === Sort@os2) ∧
   Simplify[ω1 == ω2] ∧ CF@λ1 == CF@λ2]
```

No rules for linear operations!

The crossings and the point:

```
Γ[X_{i_,j_,k_,l_}[S_, T_]] := Γ[{l, i}, {j, k}, <|ξ_i → S, x_j → T, x_k → S, ξ_l → T|>,
   T^{-1/2}, CF[{ξ_l, ξ_i}.(1 1-T; 0 T).{x_j, x_k}]];
Γ[X̄_{i_,j_,k_,l_}[S_, T_]] := Γ[{i, j}, {k, l}, <|ξ_i → S, ξ_j → T, x_k → S, x_l → T|>,
   T^{1/2}, CF[{ξ_i, ξ_j}.(T^{-1} 0; 1-T^{-1} 1).{x_k, x_l}]];
Γ[X_{i_,j_,k_,l_}] := Γ[X_{i,j,k,l}[τ_i, τ_l]];
Γ[X̄_{i_,j_,k_,l_}] := Γ[X̄_{i,j,k,l}[τ_i, τ_j]];
Γ[P_{i_,j_}[T_]] := Γ[{i}, {j}, <|ξ_i → T, x_j → T|>, 1, ξ_i x_j];
Γ[P_{i_,j_}] := Γ[P_{i,j}[τ_i]];
```

Proof. Recall that $c_{x,\xi}\colon(1,\xi,x,x\xi)w'\mapsto(1,0,0,1)w'$, write $\lambda=\mu+\eta x+\xi y+\alpha\xi x$, and ponder $\mathbb{e}^\lambda=$

$$\ldots+\frac{1}{k!}\underbrace{(\mu+\eta x+\xi y+\alpha\xi x)(\mu+\eta x+\xi y+\alpha\xi x)\cdots(\mu+\eta x+\xi y+\alpha\xi x)}_{k\text{ factors}}+\ldots.$$

Then $c_{x,\xi}\mathbb{e}^\lambda$ has three contributions:

- $\mathbb{e}^\mu$, from the term proportional to 1 (namely, independent of ξ and x) in $\mathbb{e}^\lambda$
- $-\alpha\mathbb{e}^\mu$, from the term proportional to $x\xi$, where the x and the ξ come from the same factor above.
- $\eta y\mathbb{e}^\mu$, from the term proportional to $x\xi$, where the x and the ξ come from different factors above.

So $c_{x,\xi}\mathbb{e}^\lambda=\mathbb{e}^\mu(1-\alpha+\eta y)=(1-\alpha)\mathbb{e}^\mu(1+\eta y/(1-\alpha))=(1-\alpha)\mathbb{e}^\mu\mathbb{e}^{\eta y/(1-\alpha)}=(1-\alpha)\mathbb{e}^{\mu+\eta y/(1-\alpha)}$.

□

6. An Implementation of Γ.

If I didn't implement I wouldn't believe myself.

Written in Mathematica [Wo], available as the notebook `Gamma.nb` at http://drorbn.net/mo21/ap. Code lines are highlighted in grey, demo lines are plain. We start with canonical forms for quadratics with rational function coefficients:

```
CCF[ℰ_] := Factor[ℰ];
CF[ℰ_] := Module[{vs = Union@Cases[ℰ, (ξ | x)_, ∞]},
   Total[(CCF[#[[2]]] (Times @@ vs^#[[1]])) & /@ CoefficientRules[ℰ, vs]]];
```

Contractions:

```
c_{h_,t_}@Γ[is_, os_, cs_, ω_, λ_] := Module[{α, η, y, μ},
   α = ∂_{ξ_t,x_h} λ;  μ = λ /. ξ_t | x_h → 0;
   η = ∂_{x_h} λ /. ξ_t → 0;  y = ∂_{ξ_t} λ /. x_h → 0;
   Γ[
     DeleteCases[is, t], DeleteCases[os, h], KeyDrop[cs, {x_h, ξ_t}],
     CCF[(1 - α) ω], CF[μ + η y / (1 - α)]
   ] /. If[MatchQ[cs[ξ_t], τ_], cs[ξ_t] → cs[x_h], cs[x_h] → cs[ξ_t]]];
c@Γ[is_, os_, cs_, ω_, λ_] := Fold[c_{#2,#2}[#1] &, Γ[is, os, cs, ω, λ], is ⋂ os]
```

Automatic intelligent contractions:

```
Γ[{γ_Γ}] := c[γ];
Γ[{γ1_Γ, γs__Γ}] := Module[{γ2},
  γ2 = First@MaximalBy[{γs}, Length[γ1[[1]] ⋂ #[[2]]] + Length[γ1[[2]] ⋂ #[[1]]] &];
  Γ[Join[{c[γ1 γ2]}, DeleteCases[{γs}, γ2]]] ]
Γ[𝒪s_List] := Γ[Γ /@ 𝒪s]
```

Video and more at http://www.math.toronto.edu/~drorbn/Talks/MoscowByWeb-2104/